

Der innere Aufbau Weißer Zwergsterne

Manfred Hanke (Gruppe 5: Hanke / Nieva)

20. Dez 2006

1 Zielsetzung

Es soll Dichteverlauf in einem polytropen Weißen Zwerg numerisch aufintegriert werden.

2 Methoden

2.1 Sternaufbaugleichungen

Die Gravitationskraft $dF = -\frac{Gm(r)}{r^2} dm$ regelt den Druckverlauf im Gleichgewicht:

$$\frac{dp}{dr} = \frac{dF}{dA dr} = -\frac{Gm(r)}{r^2} \frac{dm}{dV} = -\frac{Gm(r)}{r^2} \rho(r)$$

In der polytropen Näherung skaliert der Druck $p(r)$ mit Dichte $\rho(r)$ in der Potenz γ :

$$p(r) = p(0) \cdot \left(\frac{\rho(r)}{\rho(0)} \right)^\gamma \Rightarrow \frac{dp}{dr} = p(0) \cdot \frac{\gamma \cdot \rho(r)^{\gamma-1}}{\rho(0)^\gamma} \frac{d\rho}{dr} \Rightarrow \frac{dp}{dr} = \frac{\rho(0) \cdot \rho(0)^{\gamma-1}}{p(0) \cdot \gamma \cdot \rho(r)^{\gamma-1}}$$

Mit $\bar{r} = \frac{r}{R_d}$, $\bar{\rho}(\bar{r}) = \frac{\rho(r)}{\rho(0)}$ und $\bar{m}(\bar{r}) = \frac{m(r)}{M_d}$ (dimensionslos) ergibt sich durch Einsetzen:

$$\begin{aligned} \frac{d\bar{\rho}}{d\bar{r}} &= \frac{R_d}{\rho(0)} \frac{d\rho}{dr} = \frac{R_d}{\rho(0)} \frac{\rho(0) \cdot \bar{\rho}(\bar{r})^{1-\gamma}}{p(0) \cdot \gamma} \cdot \left(-\frac{G \cdot (M_d \cdot \bar{m}(\bar{r}))}{(R_d \cdot \bar{r})^2} \cdot (\rho(0) \cdot \bar{\rho}(\bar{r})) \right) \\ \Leftrightarrow \frac{d\bar{\rho}}{d\bar{r}} &= -A \frac{\bar{m}(\bar{r}) \cdot \bar{\rho}(\bar{r})^{2-\gamma}}{\bar{r}^2} \quad \text{mit} \quad A = \frac{G M_d \rho(0)}{p(0) \gamma R_d} \end{aligned}$$

Die Masse einer Kugeloberfläche ist

$$\begin{aligned} \frac{dm}{dr} &= 4\pi r^2 \rho(r) \Rightarrow \frac{d\bar{m}}{d\bar{r}} = \frac{R_d}{M_d} \frac{dm}{dr} = \frac{R_d}{M_d} 4\pi (R_d \cdot \bar{r})^2 \cdot (\rho(0) \cdot \bar{\rho}(\bar{r})) \\ \Leftrightarrow \frac{d\bar{m}}{d\bar{r}} &= B \bar{r}^2 \bar{\rho}(\bar{r}) \quad \text{mit} \quad B = \frac{4\pi R_d^3 \rho(0)}{M_d} \end{aligned}$$

Da die Dimensionsskalen R_d und M_d beliebig sind, kann $A = B = 1$ gefordert werden.

Für $\gamma = 2$ lauten die beiden Zustandsgleichungen $\frac{d\bar{\rho}}{d\bar{r}} = -\frac{\bar{m}(\bar{r})}{\bar{r}^2}$ und $\frac{d\bar{m}}{d\bar{r}} = \bar{r}^2 \bar{\rho}(\bar{r})$.

Die Lösung dazu ist $\bar{\rho}(\bar{r}) = \frac{\sin \bar{r}}{\bar{r}}$, da die Funktion $\bar{m}(\bar{r}) = -\bar{r}^2 \frac{d\bar{\rho}}{d\bar{r}} = -(\cos \bar{r} \cdot \bar{r} - \sin \bar{r})$ auch $\frac{d\bar{m}}{d\bar{r}} = \sin \bar{r} \cdot \bar{r} - \cos \bar{r} + \cos \bar{r} = \bar{r}^2 \bar{\rho}(\bar{r})$ erfüllt.

2.2 Runge-Kutta-Verfahren vierter Ordnung

Ein(e) Differentialgleichung[s-System] $\vec{y}'(x) = \vec{f}(x, \vec{y}(x))$ kann nach Runge-Kutta durch folgende numerische Integration von x_n nach $x_{n+1} = x_n + h$ approximativ gelöst werden:

$$\left. \begin{aligned} \vec{\Delta}_1 &= h \cdot \vec{f}(x_n, \vec{y}(x_n)) \\ \vec{\Delta}_2 &= h \cdot \vec{f}\left(x_n + \frac{h}{2}, \vec{y}(x_n) + \frac{\vec{\Delta}_1}{2}\right) \\ \vec{\Delta}_3 &= h \cdot \vec{f}\left(x_n + \frac{h}{2}, \vec{y}(x_n) + \frac{\vec{\Delta}_2}{2}\right) \\ \vec{\Delta}_4 &= h \cdot \vec{f}\left(x_n + h, \vec{y}(x_n) + \vec{\Delta}_3\right) \end{aligned} \right\} \Rightarrow \vec{y}(x_{n+1}) = \vec{y}(x_n) + \frac{\vec{\Delta}_1}{6} + \frac{\vec{\Delta}_2}{3} + \frac{\vec{\Delta}_3}{3} + \frac{\vec{\Delta}_4}{6} + O(h^5)$$

3 Durchführung

Um das Runge-Kutta-Verfahren in der allgemeinsten Form umzusetzen, werden Datenfelder verwendet, deren Dimension jedem Programmsegment soweit nötig übergeben wird. Die Hilfs-SUBROUTINE `ascVec(v0, a1, v1, s, n)` ('add scaled vector') berechnet komponentenweise die Summe $\vec{s} = \vec{v}_0 + a_1 \vec{v}_1$ von n -dimensionalen Vektoren.

Die SUBROUTINE `RKstep(x, y, get_f, h, n)` führt einen Runge-Kutta-Schritt der Länge h im n -dimensionalen Differentialgleichungs-System $\vec{y}'(x) = \vec{f}(x, \vec{y}(x))$ aus, wobei der Ableitungsvektor $\vec{f}$ von der übergebenen SUBROUTINE `get_f` durch `CALL get_f(x, y, f)` berechnet wird. Um die vier Ableitungen $\vec{f}_1 - \vec{f}_4$ an den Stützstellen auszurechnen, wird die Hilfsvariable y_{tmp} mittels `ascVec` auf den jeweiligen Wert gesetzt. Schließlich wird $\vec{y}$ mit den Ableitungen $f_1 - f_4$ nach der Runge-Kutta-Formel verändert und x ebenfalls um die Schrittweite h weitergesetzt.

Zur Berechnung des Sternaufbaus werden die Zustandsgrößen zu einem Vektor $\vec{y} = \begin{pmatrix} \bar{\rho} \\ \bar{m} \end{pmatrix}$ zusammengefaßt. Die Ableitungen werden nach obigen Schema in der SUBROUTINE `gpoly2(r, y, f)` für $\gamma = 2$, $A = B = 1$ bzw. SUBROUTINE `get_WD(r, y, f)` für $\gamma = \frac{5}{3}$ (γ ist in `get_WD` als ein im Prinzip beliebiger PARAMETER definiert) berechnet, d.h. es wird $f_1 := -\frac{\bar{m} \cdot \bar{\rho}^{2-\gamma}}{\bar{r}^2}$ und $f_2 := \bar{r}^2 \cdot \bar{\rho}$ gesetzt. Die formale Singularität am Startwert $\bar{r} = 0$ wird durch $f_1 := 0$ für $\bar{m} = 0$ umgangen.

Im Hauptprogramm wird jeweils das 2-dimensionale Felds $\vec{y}$ mit den Anfangswerten $\bar{\rho}(0) = 1$ und $\bar{m}(0) = 0$ bei $\bar{r} = 0$ initialisiert. Dann werden die Werte in Dateien ausgegeben und der `RKstep` mit der als EXTERNAL deklarierten SUBROUTINE `gpoly2` bzw. `get_WD` ausgeführt, solange $\bar{\rho} \geq 0$.

4 Programm

4.1 Fortran-Code

```

SUBROUTINE ascVec(v0, a1, v1, s, n)
c  add scaled (n-dimensional) vector: {s} = {v0} + a1*{v1}
  INTEGER n, i
  REAL v0(n), a1, v1(n), s(n)
  DO 100 i=1, n
100    s(i) = v0(i) + a1*v1(i)
  END

SUBROUTINE RKstep(x, y, get_f, h, n)
c  does a 4th order Runge-Kutta step for {y}'=f(x,{y}),
c  where {y} is a n-dimensional vector, from x to x+h
  INTEGER n
  REAL x, y(n), h
  REAL ytmp(n), f1(n), f2(n), f3(n), f4(n)

c  CALL get_f(x, y, f1)
  CALL ascVec(y, 0.5*h, f1, ytmp, n) ! ytmp = y + 0.5*h*f1
  CALL get_f(x+0.5*h, ytmp, f2)
  CALL ascVec(y, 0.5*h, f2, ytmp, n) ! ytmp = y + 0.5*h*f2
  CALL get_f(x+0.5*h, ytmp, f3)
  CALL ascVec(y, h, f3, ytmp, n) ! ytmp = y + h*f3
  CALL get_f(x+h, ytmp, f4)

c  CALL ascVec(y, h/6.0, f1, y, n) ! y += h/6*f1
  CALL ascVec(y, h/3.0, f2, y, n) ! y += h/3*f2
  CALL ascVec(y, h/3.0, f3, y, n) ! y += h/3*f3
  CALL ascVec(y, h/6.0, f4, y, n) ! y += h/6*f4
  x = x + h
  END

SUBROUTINE gpoly2(r, y, f)
c  get diff. eq. for polytrop star with gamma=2, A=1, B=1
  REAL r, y(2), f(2)
  IF(y(2) .NE. 0.0) THEN
    f(1) = - y(2) / (r*r)
  ELSE
c    avoid singularity at r=0
    f(1) = 0
  END IF
  f(2) = r*r * y(1)
  END
```

```

SUBROUTINE get_WD(r, y, f)
c  get differential equation for polytrop white dwarf
c  y(1) = rho => f(1) = rho' = -m * (rho**(2-gamma)) / r**2
c  y(2) = m => f(2) = m' = r**2 * rho
  REAL r, y(2), f(2)
  REAL gamma
  PARAMETER (gamma=5.0/3.0)
  IF(y(2) .NE. 0.0) THEN
    f(1) = -y(2) * (y(1)**(2-gamma)) / (r*r)
  ELSE
c    avoid singularity at r=0
    f(1) = 0
  END IF
  f(2) = r*r * y(1)
  END

PROGRAM wDwarf

c  REAL r, y(2) ! y(1) = rho, y(2) = m
  EXTERNAL gpoly2, get_WD

  r = 0
  y(1) = 1
  y(2) = 0
  OPEN(1, FILE='2_gamma2.dat')
100  WRITE(1, '(F4.2,2F10.6)') r, y(1), y(2)
  CALL RKstep(r, y, gpoly2, 0.01, 2)
  IF(y(1) .GT. 0.0) GOTO 100
  CLOSE(1)

c  r = 0
  y(1) = 1
  y(2) = 0
  OPEN(1, FILE='2_wDwarf.dat')
200  WRITE(1, '(F6.4,2F10.6)') r, y(1), y(2)
  CALL RKstep(r, y, get_WD, 0.0001, 2)
  IF(y(1) .GT. 0.0) GOTO 200
  CLOSE(1)

c  END

c  The last values of y(1) and y(2) on the surface
c  determine the dimensions R[d] and M[d].
```

4.2 Gnuplot-Script

```
set border 3
set xtics nomirror
set ytics nomirror
set term png size 360,240
```


```
set output "2-g2_rho.png"
set xlabel "r / R[d]"
set ylabel "rho(r) / rho(0)"
plot "2-gamma2.dat" not, sin(x)/x not
```

```
set output "2-g2_m.png"
set xlabel "r / R[d]"
set ylabel "m(r) / M[d]"
plot "2-gamma2.dat" u 1:3 not, sin(x)-cos(x)*x not
```

```
set output "2-WD_rho.png"
set xlabel "r / R[d]"
set ylabel "rho(r) / rho(0)"
plot [] [0:] "2-wDwarf.dat" not, sin(x)/x not
```


```
set output "2-WD_m.png"
set xlabel "r / R[d]"
set ylabel "m(r) / M[d]"
plot "2-wDwarf.dat" u 1:3 not, sin(x)-cos(x)*x not
```

5 Ergebnisse und Auswertung



Die Berechnung des Sternaufbaus für $\gamma = 2$ weicht bei der Schrittweite 0.01 um nicht mehr als $5 \cdot 10^{-6}$ oder 0.15% von der exakten Lösung $\bar{\rho}(\bar{r}) = \frac{\sin \bar{r}}{\bar{r}}$ ab. Selbst bei $h = 0.5$ ist das Verfahren noch so genau, daß die Abweichungen stets kleiner als 0.011 oder 2% sind.

Für den Weißen Zwerg mit $\gamma = \frac{5}{3} < 2$ nimmt die Dichte nicht so schnell ab:



Die Oberfläche des weien Zwergs ist durch $\bar{\rho}(\bar{R}) = 0$ definiert. $\Rightarrow \bar{R} = \frac{R_s}{R_d} = 4.4724$. Die Gesamtmasse beträgt $M_s = \bar{m}(\bar{R}) \cdot M_d = 4.9837 M_d$.

$$\text{Aus } B = \frac{4\pi R_d^3 \rho(0)}{M_d} = 1 \text{ folgt } \rho(0) = \frac{M_d}{4\pi R_d^3} = \frac{(4.4724)^3}{4.9837 \cdot 3} \cdot \frac{M_s}{\frac{4}{3}\pi R_s^3} = 5.9834 \langle \rho \rangle.$$

$$\text{Aus } A = \frac{G M_d \rho(0)}{p(0) \gamma R_d} = 1 \text{ folgt } p(0) = \frac{3 G M_d \rho(0)}{5 R_d} = \frac{3 G M_d^2}{20 \pi R_d^4} = 0.76913 \frac{G M_s^2}{R_s^4}.$$

In Einheiten von $R_\odot = 6.953 \cdot 10^8$ m und $M_\odot = 1.983 \cdot 10^{30}$ kg ist $\rho(0) = 8427 \frac{\text{kg}}{\text{m}^3} \cdot \left(\frac{M_s}{M_\odot}\right) \cdot \left(\frac{R_s}{R_\odot}\right)^{-3}$ und mit $G = 6.687 \cdot 10^{-11} \frac{\text{m}^3}{\text{kg s}^2}$ weiterhin $p(0) = 8.65 \cdot 10^{14} \text{ Pa} \cdot \left(\frac{M_s}{M_\odot}\right)^2 \cdot \left(\frac{R_s}{R_\odot}\right)^{-4}$.

Für $R_s = 0.015 R_\odot$ und $M_s = 0.6 M_\odot$ ergibt sich schließlich $\rho(0) = 1.50 \cdot 10^9 \frac{\text{kg}}{\text{m}^3}$ und $p(0) = 6.15 \cdot 10^{21} \text{ Pa}$, was gerade dem Entartungsdruck $p_{\text{Fermi}} = \frac{\hbar^2 (3\pi^2)^{2/3}}{5 m} \cdot n_{\text{El}}^{5/3}$ eines (nicht-relativistischen) Elektronengases dieser Dichte $\rho(0)$ entspricht: Die Teilchendichte n_{El} der Elektronen kann aus der Massendichte ρ bei angenommener Abgabe von Z Elektronen pro A atomaren Masseneinheiten zu $n_{\text{El}} = Z \cdot \frac{\rho}{A \cdot 1 \text{ u}} = \frac{2Z}{A} \cdot \frac{\rho}{2 \cdot 1.66 \cdot 10^{-27} \text{ kg}}$ abgeschätzt werden.

$\Rightarrow p_{\text{Fermi}} = 3.16 \cdot 10^6 \text{ Pa} \cdot \left(\frac{2Z}{A} \cdot \frac{\rho}{\text{kg m}^{-3}}\right)^{5/3}$, was das polytrope Modell mit $\gamma = \frac{5}{2}$ rechtfertigt. Mit $\langle \frac{2Z}{A} \rangle \approx 1$ und obigem $\rho(0)$ ergibt sich $p_{\text{Fermi}}(0) = 6.21 \cdot 10^{21} \text{ Pa} \approx p(0)$.

6 Ausblick: Abschätzung der Massenobergrenze

Man könnte versuchen, diese Übereinstimmung mit p_{Fermi} zum Grundprinzip zu erheben:

$$p(0) = 8.65 \cdot 10^{14} \text{ Pa} \cdot \left(\frac{M_s}{M_\odot}\right)^2 \cdot \left(\frac{R_s}{R_\odot}\right)^{-4} = 3.16 \cdot 10^6 \text{ Pa} \cdot \left(8427 \cdot \left(\frac{M_s}{M_\odot}\right) \cdot \left(\frac{R_s}{R_\odot}\right)^{-3}\right)^{5/3}$$

Daraus würde folgen, daß der Radius eines Weißen Zwergs $R_s = 0.013 R_\odot \cdot \left(\frac{M_s}{M_\odot}\right)^{-1/3}$ ist und demzufolge mit zunehmender Masse M_s abnimmt!

Alles in die Formel für die Fermi-Geschwindigkeit $v_{\text{Fermi}} = \frac{\hbar (3\pi^2 n)^{1/3}}{m}$ eingesetzt, liefert

$$v_{\text{Fermi}} = \frac{1.055 \cdot 10^{-34} \text{ Js} \cdot (3\pi^2)^{1/3}}{9.11 \cdot 10^{-31} \text{ kg}} \cdot \left(\frac{2Z}{A} \cdot \frac{8 \, 427 \, \frac{\text{kg}}{\text{m}^3} \cdot \left(\frac{M_s}{M_\odot}\right) \cdot \left(0.013 \cdot \left(\frac{M_s}{M_\odot}\right)^{-1/3}\right)^{-3}}{2 \cdot 1.66 \cdot 10^{-27} \text{ kg}} \right)^{1/3}$$

Das Endergebnis der (nicht-relativistischen) Berechnung $v_{\text{Fermi}} = 3.83 \cdot 10^8 \frac{\text{m}}{\text{s}} \cdot \left(\frac{M}{M_\odot}\right)^{2/3}$ führt mit der Forderung $v_{\text{Fermi}} < c$ zu $M_s < 0.7 M_\odot$, was einer äußerst groben Abschätzung der Massengrenze entspricht, die der Chandrasekhar-Grenze $M_s < 1.4 M_\odot$ jedoch schon erstaunlich nahe kommt.

| Mh